```
!newt2complex.tru by Amy Smith.  last modified 2/10/00.  (with parts
!from R. Neidinger's programs)
!Colors basins of roots found using Newton's method on polynomials
!of the form p(z)=(z-i)(z+i)(z-c).  Allows user to zoom in on picture.
!-------------------------------------------------------------------

print "Enter c value"
INPUT c
LET amin = min(c,0) - abs(c)/2
LET amax = max(c,0) + abs(c)/2
LET bmin = -1.5
LET bmax = 1.5
IF c = 0 then
   LET amin = -2
   LET amax = 2
END IF
print "real axis range:";amin;",";amax
print "imaginary axis range:";bmin;",";bmax
print "Resize to desired window size (smaller is faster)."
get key zzz
set window amin,amax,bmin,bmax
ask pixels apix, bpix
let numcolors = 15
let numits = 25      ! maximum number of iteration
set background color "black"
clear

LET k= 1     ! use every kth pixel, 10 for quick preview, 1 for full picture

CALL cvalue
CALL paint

SET COLOR "white"
!LIBRARY "P:\math\rineidinger\m437\graphlib"
!CALL ticks(.5,.5)
BOX CIRCLE -sqr(.1),sqr(.1),1-sqr(.1),1+sqr(.1)
BOX CIRCLE -sqr(.1),sqr(.1),-1-sqr(.1),-1+sqr(.1)
BOX CIRCLE (c-sqr(.1)),(c+sqr(.1)),-sqr(.1),sqr(.1)
SET TEXT JUSTIFY "CENTER", "HALF"
PLOT TEXT, AT c,0: STR$(c)
PLOT TEXT, AT 0,1: "i"
PLOT TEXT, AT 0,-1: "-i"

DO
   GET POINT newx,newy
   LET width = amax-amin
   LET height = bmax-bmin
   LET amin = newx - width/6
   LET amax = newx + width/6
   LET bmin = newy - height/6
   LET bmax = newy + height/6
   set window amin,amax,bmin,bmax
   ask pixels apix, bpix
   clear
   CALL paint
LOOP
```

```
!--------------------------------------------------------------------------------
SUB cvalue
   LET rna0b0 = c
   LET rna0b2 = -2*c
   LET rna0b4 = c
   LET rna1b0 = -2*c^2
   LET rna1b2 = -6+2*c^2
   LET rna1b4 = 6
   LET rna2b0 = 2*c
   LET rna2b2 = -6*c
   LET rna3b0 = 2+2*c^2
   LET rna3b2 = 12
   LET rna4b0 = -7*c
   LET rna5b0 = 6

   LET ina0b1 = 2*c^2
   LET ina0b3 = 2*(-1+c^2)
   LET ina0b5 = 6
   LET ina1b1 = -8*c
   LET ina1b3 = -8*c
   LET ina2b1 = 6+2*c^2
   LET ina2b3 = 12
   LET ina3b1 = -8*c
   LET ina4b1 = 6

   LET da0b0 = 1
   LET da0b2 = -6+4*c^2
   LET da0b4 = 9
   LET da1b0 = -4*c
   LET da1b2 = -12*c
   LET da2b0 = 6+4*c^2
   LET da2b2 = 18
   LET da3b0 = -12*c
   LET da4b0 = 9
END SUB

SUB N(a,b)

   LET a2 = a*a
   LET a3 = a*a2
   LET a4 = a*a3
   LET a5 = a*a4

   LET b2 = b*b
   LET b3 = b*b2
   LET b4 = b*b3
   LET b5 = b*b4

   LET den = da0b0 + da0b2*b2 +da0b4*b4 + (da1b0 + da1b2*b2)*a + &
&      (da2b0 + da2b2*b2)*a2 + da3b0*a3 + da4b0*a4
   LET newa = (rna0b0 + rna0b2*b2 + rna0b4*b4 + (rna1b0 + &
&      rna1b2*b2 + rna1b4*b4)*a + (rna2b0 + rna2b2*b2)*a2 + &
&      (rna3b0 + rna3b2*b2)*a3 + rna4b0*a4 + rna5b0*a5)/den
   LET newb = (ina0b1*b + ina0b3*b3 + ina0b5*b5 + (ina1b1*b + &
&      ina1b3*b3)*a + (ina2b1*b + ina2b3*b3)*a2 + ina3b1*a3*b + &
```

```
&       ina4b1*a4*b)/den
   LET a = newa
   LET b = newb
END SUB

SUB paint
   FOR a0= amin to amax step k*(amax-amin)/apix
      FOR b0= bmin to bmax step k*(bmax-bmin)/bpix
              LET a=a0
              LET b=b0
              LET n=0
              DO WHILE n<=numits and (a^2+(b-1)^2) >= .1 and &
&                       (a^2+(b+1)^2) >= .1 and ((a-c)^2+b^2) >= .1
                 CALL N(a,b)
                 LET n=n+1
              LOOP
              IF  (a^2+(b-1)^2) < .1 then    !converges to i
                      SET COLOR "green"
                      PLOT a0,b0
              ELSEIF  (a^2+(b+1)^2) < .1 then !converges to -i
                      SET COLOR "red"
                      PLOT a0,b0
              ELSEIF  ((a-c)^2+b^2) < .1 then !converges to c
                      SET COLOR "blue"
                      PLOT a0,b0
              END IF
      NEXT b0
   NEXT a0
END SUB

END
```

Coefficients of Newton's method on cubic polynomials with two complex conjugate roots and one real root

q[z_] = (z-I)(z+I)(z-c)

$(-i+z)\,(i+z)\,(-c+z)$

Nq[z_]=z-q[z]/q'[z]

$$z-\frac{(-i+z)\,(i+z)\,(-c+z)}{(-i+z)\,(i+z)+(-i+z)\,(-c+z)+(i+z)\,(-c+z)}$$

Simplify[Nq[z]]

$$\frac{c-c\,z^2+2\,z^3}{1-2\,c\,z+3\,z^2}$$

Nq[a+I b]

$$a+i\,b-\frac{(-i+a+i\,b)\,(i+a+i\,b)\,(a+i\,b-c)}{(-i+a+i\,b)\,(i+a+i\,b)+(-i+a+i\,b)\,(a+i\,b-c)+(i+a+i\,b)\,(a+i\,b-c)}$$

result=ComplexExpand[Nq[a+I b],TargetFunctions->{Re,Im}];

realPart=Simplify[result /.I→0]

$$\frac{6\,a^5-7\,a^4\,c+a^2\,(2-6\,b^2)\,c+(-1+b^2)^2\,c+2\,a\,(-1+b^2)\,(3\,b^2+c^2)+2\,a^3\,(1+6\,b^2+c^2)}{1+9\,a^4+9\,b^4-12\,a^3\,c-4\,a\,(c+3\,b^2\,c)+b^2\,(-6+4\,c^2)+a^2\,(6+18\,b^2+4\,c^2)}$$

CoefficientList[Numerator[realPart],a]

$\{(-1+b^2)^2\,c,\ 2\,(-1+b^2)\,(3\,b^2+c^2),\ (2-6\,b^2)\,c,\ 2\,(1+6\,b^2+c^2),\ -7\,c,\ 6\}$

CoefficientList[Numerator[realPart],{a,b}]

$\{\{c,0,-2\,c,0,c\},\{-2\,c^2,0,-6+2\,c^2,0,6\},$
$\{2\,c,0,-6\,c,0,0\},\{2+2\,c^2,0,12,0,0\},\{-7\,c,0,0,0,0\},\{6,0,0,0,0\}\}$

CoefficientList[Denominator[realPart],a]

$\{1+9\,b^4+b^2\,(-6+4\,c^2),\ -4\,(c+3\,b^2\,c),\ 6+18\,b^2+4\,c^2,\ -12\,c,\ 9\}$

CoefficientList[Denominator[realPart],{a,b}]

$\{\{1,0,-6+4\,c^2,0,9\},\{-4\,c,0,-12\,c,0,0\},\{6+4\,c^2,0,18,0,0\},\{-12\,c,0,0,0,0\},\{9,0,0,0,0\}\}$

imPart=Simplify[ComplexExpand[-I*result]/.I→0]

$(2\,b\,(3\,a^4+3\,b^4-4\,a^3\,c-4\,a\,(1+b^2)\,c+c^2+b^2\,(-1+c^2)+a^2\,(3+6\,b^2+c^2)))\,/$
$(1+9\,a^4+9\,b^4-12\,a^3\,c-4\,a\,(c+3\,b^2\,c)+b^2\,(-6+4\,c^2)+a^2\,(6+18\,b^2+4\,c^2))$

Denominator[imPart]===Denominator[realPart]

True

CoefficientList[Numerator[imPart],a]

$\{6\,b^5+2\,b\,c^2+2\,b^3\,(-1+c^2),\ -8\,b\,(1+b^2)\,c,\ 2\,b\,(3+6\,b^2+c^2),\ -8\,b\,c,\ 6\,b\}$

CoefficientList[Numerator[imPart],{a,b}]

$\{\{0,2\,c^2,0,2\,(-1+c^2),0,6\},\{0,-8\,c,0,-8\,c,0,0\},$
$\{0,6+2\,c^2,0,12,0,0\},\{0,-8\,c,0,0,0,0\},\{0,6,0,0,0,0\}\}$

```
!newtmult2.tru by Amy Smith.  last modified 3/28/00.  (with parts
!from R. Neidinger's programs)
!Colors basins of roots found using Newton's method on polynomials
!of the form p(z)=z^2(z-1).  Allows user to zoom in on
!picture.
!---------------------------------------------------------------------

LET amin = -.5
LET amax = 1.5
LET bmin = -.75
LET bmax = .75
print "real axis range:";amin;",";amax
print "imaginary axis range:";bmin;",";bmax
print "Resize to desired window size (smaller is faster)."
get key zzz
set window amin,amax,bmin,bmax
ask pixels apix, bpix
let numcolors = 15
let numits = 25      ! maximum number of iteration
set background color "black"
clear

LET k= 1     ! use every kth pixel, 10 for quick preview, 1 for full picture

CALL cvalue
CALL paint

SET COLOR "white"
!LIBRARY "P:\math\rineidinger\m437\graphlib"
!CALL ticks(.1,.1)
BOX CIRCLE -(1/4),(1/4),-(1/4),(1/4)
BOX CIRCLE 1-(1/4),1+(1/4),-(1/4),(1/4)
SET TEXT JUSTIFY "CENTER", "HALF"
PLOT TEXT, AT 0,0: "0"
PLOT TEXT, AT 1,0: "1"

DO
   GET POINT newx,newy
   LET width = amax-amin
   LET height = bmax-bmin
   LET amin = newx - width/6
   LET amax = newx + width/6
   LET bmin = newy - height/6
   LET bmax = newy + height/6
   set window amin,amax,bmin,bmax
   ask pixels apix, bpix
   clear
   CALL paint
LOOP

!--------------------------------------------------------------------------------

SUB cvalue

   LET rna0b2 = 1
   LET rna1b0 = 2
```

```
    LET rna1b2 = 6
    LET rna2b0 = -7
    LET rna3b0 = 6

    LET ina0b1 = 2
    LET ina0b3 = 6
    LET ina1b1 = -8
    LET ina2b1 = 6

    LET da0b0 = 4
    LET da0b2 = 9
    LET da1b0 = -12
    LET da2b0 = 9
END SUB

SUB N(a,b)
    LET a2 = a*a
    LET a3 = a*a2

    LET b2 = b*b
    LET b3 = b*b2

    LET den = da0b0 + da0b2*b2 +da1b0*a + da2b0*a2
    LET newa = (rna0b2*b2 + (rna1b0 + rna1b2*b2)*a +rna2b0*a2 + rna3b0*a3)/den
    LET newb = (ina0b1*b + ina0b3*b3 + ina1b1*b*a + ina2b1*a2*b)/den
    LET a = newa
    LET b = newb
END SUB

SUB paint
    FOR a0= amin to amax step k*(amax-amin)/apix
        FOR b0= bmin to bmax step k*(bmax-bmin)/bpix
              LET a=a0
              LET b=b0
              LET n=0
              DO WHILE n<=numits and ((a-1)^2+b^2)>=1/16 and (a^2+b^2)>=1/16
                 CALL N(a,b)
                 LET n=n+1
              LOOP
              IF  ((a-1)^2+b^2) < 1/16 then    !converges to 1
                 SET COLOR "green"
                 PLOT a0,b0
              ELSEIF  (a^2+b^2) < 1/16 then !converges to 0
                 SET COLOR "red"
                     PLOT a0,b0
              END IF
        NEXT b0
    NEXT a0
END SUB

END
```

Coefficients of Newton's method on cubic polynomials with one real root of multiplicity two and one real root of multiplicity one

q[z_] = (z-0)^2 (z-1)

$(-1+z)\,z^2$

Nq[z_]=z-q[z]/q'[z]

$$z-\frac{(-1+z)\,z^2}{2\,(-1+z)\,z+z^2}$$

Simplify[Nq[z]]

$$\frac{z\,(-1+2\,z)}{-2+3\,z}$$

Nq[a+I b]

$$a-\frac{(-1+a+\mathbb{i}\,b)\,(a+\mathbb{i}\,b)^2}{2\,(-1+a+\mathbb{i}\,b)\,(a+\mathbb{i}\,b)+(a+\mathbb{i}\,b)^2}+\mathbb{i}\,b$$

result=ComplexExpand[Nq[a+I b],TargetFunctions->{Re,Im}];

realPart=Simplify[result /.I→0]

$$\frac{2\,a-7\,a^2+6\,a^3+b^2+6\,a\,b^2}{4-12\,a+9\,a^2+9\,b^2}$$

CoefficientList[Numerator[realPart],a]

$\{b^2,\,2+6\,b^2,\,-7,\,6\}$

CoefficientList[Numerator[realPart],{a,b}]

{{0,0,1},{2,0,6},{-7,0,0},{6,0,0}}

CoefficientList[Denominator[realPart],a]

$\{4+9\,b^2,\,-12,\,9\}$

CoefficientList[Denominator[realPart],{a,b}]

{{4,0,9},{-12,0,0},{9,0,0}}

imPart=Simplify[ComplexExpand[-I*result]/.I→0]

$$\frac{2\,b\,(1-4\,a+3\,a^2+3\,b^2)}{4-12\,a+9\,a^2+9\,b^2}$$

Denominator[imPart]===Denominator[realPart]

True

CoefficientList[Numerator[imPart],a]

$\{2\,b+6\,b^3,\,-8\,b,\,6\,b\}$

CoefficientList[Numerator[imPart],{a,b}]

{{0,2,0,6},{0,-8,0,0},{0,6,0,0}}

```
!newt3real.tru by Amy Smith.  last modified 2/10/00.  (with parts
!from R. Neidinger's programs)
!Colors basins of roots found using Newton's method on polynomials
!of the form p(z)=z(z-1)(z-c) for 0<c<1.  Allows user to zoom in on
!picture.
!---------------------------------------------------------------------

print "Enter c value (0<c<1)"
INPUT c
LET amin = -.5
LET amax = 1.5
LET bmin = -.75
LET bmax = .75
print "real axis range:";amin;",";amax
print "imaginary axis range:";bmin;",";bmax
print "Resize to desired window size (smaller is faster)."
get key zzz
set window amin,amax,bmin,bmax
ask pixels apix, bpix
let numcolors = 15
let numits = 25      ! maximum number of iteration
set background color "black"
clear

LET k= 1     ! use every kth pixel, 10 for quick preview, 1 for full picture

CALL cvalue
CALL paint

SET COLOR "white"
!LIBRARY "P:\math\rineidinger\m437\graphlib"
!CALL ticks(.1,.1)
BOX CIRCLE -sqr(Ep2),sqr(Ep2),-sqr(Ep2),sqr(Ep2)
BOX CIRCLE 1-sqr(Ep2),1+sqr(Ep2),-sqr(Ep2),sqr(Ep2)
BOX CIRCLE (c-sqr(Ep2)),(c+sqr(Ep2)),-sqr(Ep2),sqr(Ep2)
SET TEXT JUSTIFY "CENTER", "HALF"
PLOT TEXT, AT c,0: STR$(c)
PLOT TEXT, AT 0,0: "0"
PLOT TEXT, AT 1,0: "1"

DO
   GET POINT newx,newy
   LET width = amax-amin
   LET height = bmax-bmin
   LET amin = newx - width/6
   LET amax = newx + width/6
   LET bmin = newy - height/6
   LET bmax = newy + height/6
   set window amin,amax,bmin,bmax
   ask pixels apix, bpix
   clear
   CALL paint
LOOP

!--------------------------------------------------------------------------------
SUB cvalue
```

```
    LET Ep2 = min(c^2/16,(1-c)^2/16)

    LET rna0b2 = c*(1+c)
    LET rna0b4 = 1+c
    LET rna1b2 = 2-2*c+2*c^2
    LET rna1b4 = 6
    LET rna2b0 = -c*(1+c)
    LET rna2b2 = -6*(1+c)
    LET rna3b0 = 2+6*c+2*c^2
    LET rna3b2 = 12
    LET rna4b0 = -7*(1+c)
    LET rna5b0 = 6

    LET ina0b3 = 2+2*c+2*c^2
    LET ina0b5 = 6
    LET ina1b1 = -2*c*(1+c)
    LET ina1b3 = -8*(1+c)
    LET ina2b1 = 2+10*c+2*c^2
    LET ina2b3 = 12
    LET ina3b1 = -8*(1+c)
    LET ina4b1 = 6

    LET da0b0 = c^2
    LET da0b2 = 2*(2+c+2*c^2)
    LET da0b4 = 9
    LET da1b0 = -4*c*(1+c)
    LET da1b2 = -12*(1+c)
    LET da2b0 = 4+14*c+4*c^2
    LET da2b2 = 18
    LET da3b0 = -12*(1+c)
    LET da4b0 = 9
END SUB

SUB N(a,b)
    LET a2 = a*a
    LET a3 = a*a2
    LET a4 = a*a3
    LET a5 = a*a4

    LET b2 = b*b
    LET b3 = b*b2
    LET b4 = b*b3
    LET b5 = b*b4

    LET den = da0b0 + da0b2*b2 +da0b4*b4 + (da1b0 + da1b2*b2)*a + &
&       (da2b0 + da2b2*b2)*a2 + da3b0*a3 + da4b0*a4
    LET newa = (rna0b2*b2 + rna0b4*b4 + (rna1b2*b2 + rna1b4*b4)*a &
&       + (rna2b0 + rna2b2*b2)*a2 + (rna3b0 + rna3b2*b2)*a3 + &
&       rna4b0*a4 + rna5b0*a5)/den
    LET newb = (ina0b3*b3 + ina0b5*b5 + (ina1b1*b + ina1b3*b3)*a + &
&       (ina2b1*b + ina2b3*b3)*a2 + ina3b1*a3*b + ina4b1*a4*b)/den
    LET a = newa
    LET b = newb
END SUB

SUB paint
```

```
   FOR a0= amin to amax step k*(amax-amin)/apix
      FOR b0= bmin to bmax step k*(bmax-bmin)/bpix
             LET a=a0
             LET b=b0
             LET n=0
             DO WHILE n<=numits and ((a-1)^2+b^2) >= Ep2 and &
&                    (a^2+b^2) >= Ep2 and ((a-c)^2+b^2) >= Ep2
                CALL N(a,b)
                LET n=n+1
             LOOP
             IF  ((a-1)^2+b^2) < Ep2 then    !converges to 1
                SET COLOR "green"
                PLOT a0,b0
             ELSEIF  (a^2+b^2) < Ep2 then !converges to 0
                SET COLOR "red"
                     PLOT a0,b0
             ELSEIF  ((a-c)^2+b^2) < Ep2 then !converges to c
                     SET COLOR "blue"
                     PLOT a0,b0
             END IF
      NEXT b0
   NEXT a0
END SUB

END
```

Coefficients of Newton's method on cubic polynomials with three distinct real roots

q[z_]=(z-0)(z-1)(z-c)

(-1+z) z (-c+z)

Nq[z_]=z-q[z]/q'[z]

$$z-\frac{(-1+z)\,z\,(-c+z)}{(-1+z)\,z+(-1+z)\,(-c+z)+z\,(-c+z)}$$

Simplify[Nq[z]]

$$\frac{(1+c-2z)\,z^2}{(2-3z)\,z+c\,(-1+2z)}$$

Nq[a+I b]

$$a+\mathrm{i}\,b-\frac{(-1+a+\mathrm{i}\,b)\,(a+\mathrm{i}\,b)\,(a+\mathrm{i}\,b-c)}{(-1+a+\mathrm{i}\,b)\,(a+\mathrm{i}\,b)+(-1+a+\mathrm{i}\,b)\,(a+\mathrm{i}\,b-c)+(a+\mathrm{i}\,b)\,(a+\mathrm{i}\,b-c)}$$

result=ComplexExpand[Nq[a+I b],TargetFunctions->{Re,Im}];

realPart=Simplify[result /.I→0]

$$\frac{6a^5-7a^4(1+c)+b^2(1+c)(b^2+c)-a^2(1+c)(6b^2+c)+2ab^2(1+3b^2-c+c^2)+2a^3(1+6b^2+3c+c^2)}{9a^4+9b^4+c^2-12a^3(1+c)-4a(1+c)(3b^2+c)+2b^2(2+c+2c^2)+2a^2(2+9b^2+7c+2c^2)}$$

CoefficientList[Numerator[realPart],a]

$$\{b^2(1+c)(b^2+c),\ 2b^2(1+3b^2-c+c^2),\ (-1-c)(6b^2+c),\ 2(1+6b^2+3c+c^2),\ -7(1+c),\ 6\}$$

CoefficientList[Numerator[realPart],{a,b}]

$$\{\{0,0,c(1+c),0,1+c\},\{0,0,2-2c+2c^2,0,6\},\{-c(1+c),0,-6(1+c),0,0\},\{2+6c+2c^2,0,12,0,0\},\{-7(1+c),0,0,0,0\},\{6,0,0,0,0\}\}$$

CoefficientList[Denominator[realPart],a]

$$\{9b^4+c^2+2b^2(2+c+2c^2),\ -4(1+c)(3b^2+c),\ 2(2+9b^2+7c+2c^2),\ -12(1+c),\ 9\}$$

CoefficientList[Denominator[realPart],{a,b}]

$$\{\{c^2,0,2(2+c+2c^2),0,9\},\{-4c(1+c),0,-12(1+c),0,0\},\{4+14c+4c^2,0,18,0,0\},\{-12(1+c),0,0,0,0\},\{9,0,0,0,0\}\}$$

imPart=Simplify[ComplexExpand[-I*result]/.I→0]

$$\frac{2b\,(3a^4-4a^3(1+c)-a(1+c)(4b^2+c)+b^2(1+3b^2+c+c^2)+a^2(1+6b^2+5c+c^2))}{9a^4+9b^4+c^2-12a^3(1+c)-4a(1+c)(3b^2+c)+2b^2(2+c+2c^2)+2a^2(2+9b^2+7c+2c^2)}$$

Denominator[imPart]===Denominator[realPart]

True

CoefficientList[Numerator[imPart],a]

$$\{2b^3(1+3b^2+c+c^2),\ -2b(1+c)(4b^2+c),\ 2b(1+6b^2+5c+c^2),\ -8b(1+c),\ 6b\}$$

CoefficientList[Numerator[imPart],{a,b}]

$$\{\{0,0,0,2+2c+2c^2,0,6\},\{0,-2c(1+c),0,-8(1+c),0,0\},\{0,2+10c+2c^2,0,12,0,0\},\{0,-8(1+c),0,0,0,0\},\{0,6,0,0,0,0\}\}$$

```
!parameteraxis.tru by Amy Smith.  Last modified 2/14/00.  Tests polynomials
!of the form p(z)=(z-i)(z+i)(z-c) where c is real.  Given a c value,
!the critical point of the Newton's method function, c/3, is used as an
!initial point in Newton's method. The program colors the c value green if
!the method converges to c or keeps the c value black if it does not converge.
!-------------------------------------------------------------------------------

print "Enter real axis range"
INPUT cmin
INPUT cmax
LET bmin = -1.5
LET bmax = 1.5
set window cmin,cmax,bmin,bmax
ask pixels cpix, bpix
set background color "black"
clear

  FOR c0= cmin to cmax step (cmax-cmin)/cpix
        LET c=c0
        LET a=c/3
    LET n=0
    DO WHILE n<=25 and (a-c)^2 >= .1
          LET newa = a-((a^2+1)*(a-c)/(3*a^2-2*c*a+1))
          LET a=newa
          LET n=n+1
    LOOP
  IF (a-c)^2 < .1 then !converges to c
              SET COLOR "green"
                     PLOT c0,bmin;c0,bmax
  END IF
  NEXT c0

  SET COLOR "white"
  LIBRARY "P:\math\rineidinger\m437\graphlib"
  CALL ticks(1,1)
  !print "green converges to c, black does not converge"


END
```

!bifuraxis.tru. Created by Amy Smith 4/9/00. Shows a
!bifurcation for c/3 given a range of real c values for
!Newton's method on polynomials of the form
!q(z)=(z-i)(z+i)(z-c).
!--

```
print "Enter real axis range"
INPUT cmin
INPUT cmax
LET ymin = -(max(abs(cmin),abs(cmax)))
LET ymax = abs(max(cmin,cmax))
!let ymin = -1.3
!let ymax = 1.2
!let ymin = -.5
!let ymax = 2.0
set window cmin,cmax,ymin,ymax
ask pixels cpix, ypix
!set back "black"
set back "white"
clear

CALL paint

DO
   GET POINT cmin,trashy1
   GET POINT cmax,trashy2
   GET POINT trashc1,ymin
   GET POINT trashc2,ymax
   set window cmin,cmax,ymin,ymax
   ask pixels cpix,ypix
   clear
   CALL paint
LOOP

!--------------------------------------------------

SUB paint
   FOR c0= cmin to cmax step (cmax-cmin)/cpix
         LET c=c0
         LET a=c/3
     LET n=0
!        set color "white"
!        plot c0,a
     DO WHILE n<=100
           LET newa = a-((a^2+1)*(a-c)/(3*a^2-2*c*a+1))
           LET a=newa
           LET n=n+1
     LOOP
         LET n=1
         DO WHILE n<=100
           LET newa = a-((a^2+1)*(a-c)/(3*a^2-2*c*a+1))
           LET a=newa
           !SET COLOR n
           PLOT c,a
           LET n=n+1
     LOOP
```

```
    NEXT c0


!SET COLOR "white"
set color "black"
LIBRARY "P:\math\rineidinger\m437\graphlib"
CALL ticks(1,1)
!CALL ticks(.1,.1)
!print "c-axis from";cmin;"to";cmax

END SUB

END
```